
The role of the parser

code
source tokens

errors

scanner parser IR

A parser
performs context-free syntax analysis
guides context-sensitive analysis
constructs an intermediate representation
produces meaningful error messages
attempts error correction

For the next several classes, we will look at parser construction

Syntax analysis by using a CFG

Context-free syntax is specified with a context-free grammar.
Formally, a CFG G is a 4-tuple (Vt,Vn,S,P), where:

Vt is the set of terminal symbols in the grammar.
For our purposes, Vt is the set of tokens returned by the
scanner.

Vn, the nonterminals, is a set of syntactic variables that
denote sets of (sub)strings occurring in the language.

S is a distinguished nonterminal (S ∈ Vn) denoting the entire
set of strings in L(G).
This is sometimes called a goal symbol.

P is a finite set of productions
Each production must have a single non-terminal on its
left hand side.

The set V = Vt ∪Vn is called the vocabulary of G.

Notation and terminology

a,b,c, . . . ∈ Vt

A,B,C, . . . ∈ Vn

U,V,W, . . . ∈ V

α,β ,γ, . . . ∈ V∗

u,v,w, . . . ∈ V∗
t

If A → γ then αAβ ⇒ αγβ is a single-step derivation using A → γ

Similarly, ⇒∗ and ⇒+ denote derivations of ≥ 0 and ≥ 1 steps

If S ⇒∗ β then β is said to be a sentential form of G

L(G) = {w ∈ V∗
t | S ⇒+ w}, w ∈ L(G) is called a sentence of G

Note, L(G) = {β ∈ V∗ | S ⇒∗ β}∩V∗
t

Syntax analysis

Grammars are often written in Backus-Naur form (BNF).
Example:

1 ⟨goal⟩ ::= ⟨expr⟩
2 ⟨expr⟩ ::= ⟨expr⟩⟨op⟩⟨expr⟩
3 | num
4 | id
5 ⟨op⟩ ::= +
6 | −
7 | ∗
8 | /

This describes simple expressions over numbers and identifiers.
In a BNF for a grammar, we represent

1 non-terminals with angle brackets or capital letters
2 terminals with typewriter font or underline
3 productions as in the example

Derivations

We can view the productions of a CFG as rewriting rules.
Using our example CFG (for x + 2 ∗ y):

⟨goal⟩ ⇒ ⟨expr⟩
⇒ ⟨expr⟩⟨op⟩⟨expr⟩
⇒ ⟨id,x⟩⟨op⟩⟨expr⟩
⇒ ⟨id,x⟩+ ⟨expr⟩
⇒ ⟨id,x⟩+ ⟨expr⟩⟨op⟩⟨expr⟩
⇒ ⟨id,x⟩+ ⟨num,2⟩⟨op⟩⟨expr⟩
⇒ ⟨id,x⟩+ ⟨num,2⟩ ∗ ⟨expr⟩
⇒ ⟨id,x⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩

We have derived the sentence x + 2 ∗ y.
We denote this ⟨goal⟩⇒∗ id + num ∗ id.

Such a sequence of rewrites is a derivation or a parse.

The process of discovering a derivation is called parsing.

Derivations

At each step, we chose a non-terminal to replace.
This choice can lead to different derivations.
Two are of particular interest:

leftmost derivation
the leftmost non-terminal is replaced at each step
rightmost derivation
the rightmost non-terminal is replaced at each step

The previous example was a leftmost derivation.

Rightmost derivation

For the string x + 2 ∗ y:

⟨goal⟩ ⇒ ⟨expr⟩
⇒ ⟨expr⟩⟨op⟩⟨expr⟩
⇒ ⟨expr⟩⟨op⟩⟨id,y⟩
⇒ ⟨expr⟩ ∗ ⟨id,y⟩
⇒ ⟨expr⟩⟨op⟩⟨expr⟩ ∗ ⟨id,y⟩
⇒ ⟨expr⟩⟨op⟩⟨num,2⟩ ∗ ⟨id,y⟩
⇒ ⟨expr⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩
⇒ ⟨id,x⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩

Again, ⟨goal⟩⇒∗ id + num ∗ id.

Precedence: the tree from the rightmost derivation

goal

expr

expr

expr

⟨id,x⟩

op

+

expr

⟨num,2⟩

op

*

expr

⟨id,y⟩

Treewalk evaluation computes (x + 2) ∗ y
— the “wrong” answer!
Should be x + (2 ∗ y)

Precedence: the tree from the leftmost derivation

goal

expr

expr

⟨id,x⟩

op

+

expr

expr

⟨num,2⟩

op

*

expr

⟨id,y⟩

Treewalk evaluation computes x + (2 ∗ y)
— the answer we wanted.
Same grammar, same string, two different trees.

Precedence

These two derivations point out a problem with the grammar.

It has no notion of precedence, or implied order of evaluation.

The grammar is ambiguous, as a string in the language can have
multiple parse trees.

Is precedence the only source of ambiguity? Other examples of strings
with multiple parse trees?

Ambiguity - Associativity

The expression a-b-c may be parsed as:
(a-b)-c or
a-(b-c)

In C, assignment = is right-associative. a=b=c may be parsed as:
a=(b=c) or
(a=b)=c

Removing Ambiguity

To remove ambiguity, the grammar needs to be modified:

1 ⟨goal⟩ ::= ⟨expr⟩
2 ⟨expr⟩ ::= ⟨expr⟩+ ⟨term⟩
3 | ⟨expr⟩−⟨term⟩
4 | ⟨term⟩
5 ⟨term⟩ ::= ⟨term⟩ ∗ ⟨factor⟩
6 | ⟨term⟩/⟨factor⟩
7 | ⟨factor⟩
8 ⟨factor⟩ ::= num
9 | id

This grammar enforces a precedence and associativity on the
derivation:

terms must be derived from expressions
forces the “correct” tree

Precedence

Now, for the string x + 2 ∗ y:

⟨goal⟩ ⇒ ⟨expr⟩
⇒ ⟨expr⟩+ ⟨term⟩
⇒ ⟨expr⟩+ ⟨term⟩ ∗ ⟨factor⟩
⇒ ⟨expr⟩+ ⟨term⟩ ∗ ⟨id,y⟩
⇒ ⟨expr⟩+ ⟨factor⟩ ∗ ⟨id,y⟩
⇒ ⟨expr⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩
⇒ ⟨term⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩
⇒ ⟨factor⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩
⇒ ⟨id,x⟩+ ⟨num,2⟩ ∗ ⟨id,y⟩

Again, ⟨goal⟩⇒∗ id + num ∗ id, but this time, we build the desired tree.

Precedence

goal

expr

expr

term

factor

⟨id,x⟩

+ term

term

factor

⟨num,2⟩

* factor

⟨id,y⟩

Treewalk evaluation computes x + (2 ∗ y)

Role of CFGs in Compilers

CFGs offer significant advantages for language designers, compiler
developers, and end-users of the compiler:

A grammar gives a formal, precise, yet easy-to-understand
syntactic specification of the programming languages. Useful for
end-users
For certain classes of grammars, there are procedures to
automatically construct efficient parsers from the grammar
description. Useful for compiler developers
A grammar can reveal syntactic ambiguities and trouble spots.
Useful for language designers
A grammar imparts structure to a program, which is directly used
for its translation into object code. Useful for compiler developers
A grammar allows a language to be evolved iteratively by adding
new constructs. Useful for language designers and compiler
developers

Ambiguity

If a grammar has more than one parse tree for a single sentence, then
it is ambiguous

(equivalently: more than one leftmost derivation)
Example:
⟨stmt⟩ ::= if ⟨expr⟩then ⟨stmt⟩

| if ⟨expr⟩then ⟨stmt⟩else ⟨stmt⟩
| other

Consider deriving the sentential form:
if E1 then if E2 then S1 else S2

This is the classic dangling else problem.

The ambiguity is purely grammatical: it is visible in the productions
alone, with no reference to types or declarations. A context-free
ambiguity.

Ambiguity: two trees for the same string

else matched with the inner if
stmt

if E1 then stmt

if E2 then S1 else S2

else matched with the outer if
stmt

if E1 then stmt else S2

if E2 then S1

Both yield if E1 then if E2 then S1 else S2. Which if owns the
else?

Ambiguity

We would like to parse if-then-else statements using the following
rule:

match each else with the closest unmatched then

Grammar which eliminates the ambiguity by following the above rule:
⟨stmt⟩ ::= ⟨matched⟩

| ⟨unmatched⟩
⟨matched⟩ ::= if ⟨expr⟩ then ⟨matched⟩ else ⟨matched⟩

| other
⟨unmatched⟩ ::= if ⟨expr⟩ then ⟨stmt⟩

| if ⟨expr⟩ then ⟨matched⟩ else ⟨unmatched⟩

Did the rewrite lose anything?

Not a feeling. The precise question: do the two grammars generate the
same language?

Here, yes. Both grammars are small, and it can be proved. A proof
checked by Lean 4 sits next to this deck in the course repository:
https://github.com/fplaunchpad/cs3300_m26/tree/main/lectures/03_parsing

In general, whether two CFGs generate the same language is
undecidable.

That means no algorithm answers it for every input. It does not mean
any particular case is hard, and it does not mean there is no proof.

Ambiguity

Ambiguity is often due to confusion in the context-free specification.

Context-sensitive confusions can arise from overloading.

Example:
a = b + c

In many languages, + can refer to both integer addition and floating
point addition.

Disambiguating this statement requires context:
need values of declarations
not context-free
really an issue of type

Rather than complicate parsing, we will handle this separately.

Building and using the LL(1) table
When LL(1) fails

Scanning vs. parsing

Where do we draw the line?

⟨id⟩ ::= [a−zA−z]([a−zA−z] | [0−9])∗

⟨num⟩ ::= 0 | [1−9][0−9]∗

⟨op⟩ ::= + | − | ∗ | /
⟨expr⟩ ::= ⟨expr⟩⟨op⟩⟨expr⟩ | ⟨id⟩ | ⟨digit⟩

Regular expressions are used to classify:
identifiers, numbers, keywords
REs are more concise and simpler for tokens than a grammar
more efficient scanners can be built from REs (DFAs) than
grammars

Scanning vs. parsing

Context-free grammars are used to count:
brackets: (), begin. . .end, if. . .then. . .else
imparting structure

arithmetic expressions can be described by regular expressions
but, must deal with precedence and associativity separately . . .

Syntactic analysis is complicated enough: grammar for C has around
200 productions.

Factoring out lexical analysis as a separate phase makes compiler
more manageable.

Parsing: the big picture

parser

generator

code

parser

tokens

IR

grammar

Our goal is a flexible parser generator system

Different ways of parsing: Top-down Vs Bottom-up

Top-down parsers
start at the root of derivation tree and fill in
picks a production and tries to match the input
may require backtracking
some grammars are backtrack-free (predictive)

Bottom-up parsers
start at the leaves and fill in
start in a state valid for legal first tokens
as input is consumed, change state to encode possibilities
(recognize valid prefixes)
use a stack to store both state and sentential forms

Top-down parsing

A top-down parser starts with the root of the parse tree, labelled with
the start or goal symbol of the grammar.

To build a parse, it repeats the following steps until the fringe of the
parse tree matches the input string

1 At a node labelled A, select a production A → α and construct the
appropriate child for each symbol of α

2 When a terminal is added to the fringe that doesn’t match the
input string, backtrack

3 Find next node to be expanded (must have a label in Vn)

The key is selecting the right production in step 1.

Example

⟨term⟩ ::= id | num
⟨op⟩ ::= + | −
⟨expr⟩ ::= ⟨expr⟩⟨op⟩⟨term⟩ | ⟨term⟩

Consider the string x+5.

Example: parsing x+5

⟨term⟩ ::= id | num
⟨op⟩ ::= + | −
⟨expr⟩ ::= ⟨expr⟩⟨op⟩⟨term⟩

| ⟨term⟩

Input: x + 5

1 ⟨expr⟩ ::= ⟨expr⟩⟨op⟩⟨term⟩
2 ⟨expr⟩ ::= ⟨term⟩
3 ⟨term⟩ ::= id, match x

4 ⟨op⟩ ::= +, match +

5 ⟨term⟩ ::= num, match 5

expr

expr

term

⟨id,x⟩

op

+

term

⟨num,5⟩

Example: the other choice at step 2

Step 2 could equally have
taken
⟨expr⟩ ::= ⟨expr⟩⟨op⟩⟨term⟩ a
second time.
Each application puts ⟨expr⟩
back at the left end of the
fringe.
So no terminal ever reaches
the front, no input is
consumed, and the leftmost
leaf in Vn is always ⟨expr⟩.
Nothing is matched, so nothing
fails, so step 2 never
backtracks.

expr

expr

expr

.

.

.

op term

op term

Immediate Left-recursion

Top-down parsers cannot handle left-recursion in a grammar.

Formally, a grammar is left-recursive if
∃A ∈ Vn such that A ⇒+ Aα for some string α

A grammar is said to be immediate left-recursive if
∃A ∈ Vn such that A → Aα for some string α

Our simple expression grammar is immediate left-recursive.

Eliminating immediate left-recursion

To remove immediate left-recursion, we can transform the grammar
Consider the grammar fragment:

⟨foo⟩ ::= ⟨foo⟩α
| β

where α and β do not start with ⟨foo⟩
We can rewrite this as:

⟨foo⟩ ::= β ⟨bar⟩
⟨bar⟩ ::= α⟨bar⟩

| ε

where ⟨bar⟩ is a new non-terminal

This fragment contains no immediate left-recursion

Eliminating immediate left-recursion

In general, if the grammar contains the following production rules:

⟨A⟩ ::= ⟨A⟩α1 | ⟨A⟩α2 | . . . | ⟨A⟩αm | β1 | β2 | . . . | βn

they can be replaced by the following:

⟨A⟩ ::= β1⟨A’⟩ | β2⟨A’⟩ | . . . | βn⟨A’⟩
⟨A’⟩ ::= α1⟨A’⟩ | α2⟨A’⟩ | . . .αm⟨A’⟩ | ε

Example

Consider the simplified expression grammar:

E ::= E+T | T
T ::= id | num

After eliminating left-recursion:

E ::= TE′

E′ ::= +TE′ | ε

T ::= id | num

Looking ahead to drive the choice of productions: x + 5.

How much lookahead is needed?

We saw that top-down parsers need to select a production rule at
every step, for which we may have to look ahead in the input string.

Do we need arbitrary lookahead to parse CFGs?
in general, yes
use the Earley or CYK algorithms

Fortunately
large subclasses of CFGs can be parsed with limited lookahead
most programming language constructs can be expressed in a
grammar that falls in these subclasses

Among the interesting subclasses are:
LL(1): left to right scan, left-most derivation, 1-token lookahead;

and
LR(1): left to right scan, reversed right-most derivation, 1-token

lookahead

Recursive descent parsing and Predictive parsing

If top-down parsing is performed recursively, it is also called
recursive descent parsing.

To prevent infinite recursion, the grammar should not be
left-recursive.
In general, may require backtracking if the wrong production rule is
picked.

Top-down parsing with lookahead which ensures that the correct
production rule is always picked is called predictive parsing.

Recursive descent parsing

A set of procedures, one for each non-terminal.

1 int A()
2 begin
3 foreach production of the form A → X1X2X3 · · ·Xk do
4 for i = 1 to k do
5 if Xi is a non-terminal then
6 if (Xi() = 0) then
7 backtrack; break; // Try the next production

8 else if Xi matches the current input symbol a then
9 advance the input to the next symbol;

10 else
11 backtrack; break; // Try the next production

12 if i = k+1 then
13 return 1; // Success

14 return 0; // Failure

Recursive descent parsing

Backtracks in general – in practise may not do much.
How to backtrack?
A left recursive grammar will lead to infinite loop.

Predictive parsing

Basic idea:
For any two productions A → α | β , we would like a distinct way of
choosing the correct production to expand.
For some RHS α ∈ G, define FIRST(α) as the set of tokens that
appear first in some string derived from α.

That is, for some a ∈ Vt, a ∈ FIRST(α) iff. α ⇒∗ aγ.

Key property:
Whenever two productions A → α and A → β both appear in the
grammar, we would like:

FIRST(α)∩ FIRST(β ) = φ

This would allow the parser to make a correct choice with a
lookahead of only one symbol!

Issue:
If the grammar has two productions rules of the form
A → αβ1 | αβ2, we cannot directly use predictive parsing.

Left factoring

Some grammars can be transformed by left-factoring to enable
predictive parsing.

For each non-terminal A find the longest prefix
α common to two or more of its production rules.

if α ̸= ε then replace all of the A productions
A → αβ1 | αβ2 | · · · | αβn

with
A → αA′

A′ → β1 | β2 | · · · | βn

where A′ is a new non-terminal.

Repeat until no two alternatives for a single
non-terminal have a common prefix.

Example

There are two non-terminals
to left factor:
⟨expr⟩ ::= ⟨term⟩+ ⟨expr⟩

| ⟨term⟩−⟨expr⟩
| ⟨term⟩

⟨term⟩ ::= ⟨factor⟩ ∗ ⟨term⟩
| ⟨factor⟩/⟨term⟩
| ⟨factor⟩

Question: What’s different
here from the previous
similar grammar that we’ve
seen?

Applying the transformation:

⟨expr⟩ ::= ⟨term⟩⟨expr′⟩
⟨expr′⟩ ::= +⟨expr⟩

| −⟨expr⟩
| ε

⟨term⟩ ::= ⟨factor⟩⟨term′⟩
⟨term′⟩ ::= ∗⟨term⟩

| /⟨term⟩
| ε

Left-recursion Elimination

Predictive Parsing is a form of recursive-descent parsing, and
hence cannot handle grammars with left recursion.

We have seen how to eliminate immediate left-recursion, i.e.
when there is a production rule of the form A → Aα.
However, left-recursion can also be indirect.

Example: A → Bα and B → Aβ .

In the general case, A grammar is left-recursive if ∃A ∈ Vn such
that A ⇒+ Aα for some string α.

Indirect Left-recursion Elimination

Given a CFG, to eliminate left-recursion:
1 Input: Grammar G with no cycles (no A ⇒∗ A) and no ε

productions.
2 Output: Equivalent grammar with no left-recursion.
3 begin
4 Arrange the non terminals in some order A1,A2, · · ·An;
5 foreach i = 1 · · ·n do
6 foreach j = 1 · · · i−1 do
7 For production p of the form Ai → Ajγ and

Aj → δ1|δ2| · · · |δk;
8 Replace the production p by:
9 Ai → δ1γ|δ2γ| · · · δnγ;

10 Eliminate immediate left recursion in Ai;

Example: eliminating indirect left recursion

Consider the grammar
⟨S⟩ ::= ⟨A⟩a | b
⟨A⟩ ::= ⟨S⟩d | c

with indirect left recursion ⟨S⟩ ⇒∗ ⟨S⟩da.

1. Order the nonterminals: A1 = ⟨S⟩, A2 = ⟨A⟩. For i = 1 there is no earlier
nonterminal to substitute.

2. Process i = 2, j = 1: substitute ⟨S⟩ ::= ⟨A⟩a | b into ⟨A⟩ ::= ⟨S⟩d | c:

⟨A⟩ ::= ⟨A⟩ad | bd | c

The indirect recursion is now immediate.

3. Eliminate the immediate left recursion in ⟨A⟩ (α = ad, β1 = bd, β2 = c):

⟨S⟩ ::= ⟨A⟩a | b
⟨A⟩ ::= bd⟨A’⟩ | c⟨A’⟩
⟨A’⟩ ::= ad⟨A’⟩ | ε

Indirect Left-recursion Elimination Algorithm Analysis

At the end of ith iteration of the outer loop, the algorithm ensures
that in all productions of the form Ai → Ajγ, i < j.
The algorithm assumes that the grammar has no cycles, i.e.
A ⇒∗ A is not possible for any non-terminal A.

Questions to ponder:
What happens if there are cycles in the input grammar?
What happens if there are ε-productions in the input grammar?

With a cycle, substitution can produce A → A, and immediate
left-recursion elimination does not work on it: the rewrite assumes
A → Aα with α ̸= ε. Remove cycles first.

Why ε-productions break the algorithm

Order the nonterminals as A1 = A, A2 = B, A3 = C:

A → Ca | b
B → ε

C → BA

For i = 3, j = 1: C → BA does not begin with A1 = A, so the algorithm makes
no substitution.

Next, j = 2: substitute B → ε:

C → εA = A.

But the loop does not revisit j = 1. Therefore

A ⇒ Ca ⇒ Aa,

and the indirect left recursion remains. Eliminate ε-productions before
running the algorithm.

Which grammars does this leave out?

Both preconditions rule out grammars, so it is worth asking what they
rule out in terms of languages.

Does the algorithm work for all context-free languages?
Yes, it works for all CFL which do not contain ε. For any such CFL,
we can always obtain a CFG which does not contain ε-productions
and unit-productions.

Generality

Question:
By left factoring and eliminating left-recursion, can we trans-
form an arbitrary context-free grammar to a form where it can
be predictively parsed with a single token lookahead?

Answer: No. Example:

{an0bn | n ≥ 1}∪{an1b2n | n ≥ 1}

A CFG is:
S → A | B A → aAb | a0b

B → aBbb | a1bb

Not LL(k), for any fixed k: while choosing S → A or S → B,
lookahead may contain only a’s.

Where we are

1 Context-free grammars

2 Top-down parsing
From scanning to parsing
Left recursion
Recursive descent and predictive parsing
FIRST, FOLLOW and LL(1) grammars
Building and using the LL(1) table
When LL(1) fails

Non-recursive predictive parsing

Now, a predictive parser looks like:

scanner
table-driven

parser
IR

parsing

tables

stack

source

code

tokens

Rather than writing recursive code, we build tables.
Building tables can be automated easily.

Table-driven parsers

A parser generator system often looks like:

scanner
table-driven

parser
IR

parsing

tables

stack

source

code

tokens

parser

generator
grammar

We will first look at the information required for generating the
parsing table.

FIRST

For a string of grammar symbols α, define FIRST(α) as:
the set of terminals that begin strings derived from α:
{a ∈ Vt | α ⇒∗ aβ}
If α ⇒∗ ε then ε ∈ FIRST(α)

To build FIRST(X):
1 If X ∈ Vt then FIRST(X) is {X}
2 If X → ε then add ε to FIRST(X)
3 If X → Y1Y2 · · ·Yk:

1 Put FIRST(Y1)−{ε} in FIRST(X)
2 ∀i : 1 < i ≤ k, if ε ∈ FIRST(Y1)∩·· ·∩ FIRST(Yi−1)

(i.e., Y1 · · ·Yi−1 ⇒∗ ε)
then put FIRST(Yi)−{ε} in FIRST(X)

3 If ε ∈ FIRST(Y1)∩·· ·∩ FIRST(Yk) then put ε in FIRST(X)
Repeat until no more additions can be made.

FOLLOW

For a non-terminal A, define FOLLOW(A) as
the set of terminals that can appear immediately to the right of
A in some sentential form

We choose A → α by what α starts with. A production that produces
nothing has no such symbol:

A → α look inside: FIRST(α)
A → ε look after: FOLLOW(A)

Why FOLLOW: an example

E → T E′ T → F T ′

E′ →+E | ε T ′ →∗T | /T | ε

We are at T ′ and the lookahead is +:

T ′ →∗T needs ∗
T ′ → /T needs /
T ′ → ε needs nothing

T ′ ends a T, a T is followed by E′, and E′ can start with +. So
+ ∈ FOLLOW(T ′): T ′ vanishes here, and the + is matched by E′ further
out.

Building FOLLOW

A terminal symbol has no FOLLOW set.

To build FOLLOW(A):
1 Put $ in FOLLOW(⟨goal⟩)
2 If A → αBβ :

1 Put FIRST(β )−{ε} in FOLLOW(B)
2 If β = ε (i.e., A → αB) or ε ∈ FIRST(β ) (i.e., β ⇒∗ ε) then put

FOLLOW(A) in FOLLOW(B)
Repeat until no more additions can be made

LL(1) grammars

Previous definition (slide 44)
A grammar G is LL(1) iff. for all non-terminals A, each dis-
tinct pair of productions A → β and A → γ satisfy the condition
FIRST(β )

⋂
FIRST(γ) = φ .

What if ε ∈ FIRST(β )?

Consider that the current input symbol is a. Choosing A → β consumes
nothing, so it is the right move exactly when a can legally come after A,
i.e. when a ∈ FOLLOW(A). That introduces ambiguity between
choosing:

A → β when a ∈ FOLLOW(A)

A → γ when a ∈ FIRST(γ)

Ambiguity is bad because we may need to backtrack: not predictive
parsing anymore!

LL(1) grammars

Revised definition
A grammar G is LL(1) iff. for each set of productions A → α1 |
α2 | · · · | αn:

1 FIRST(α1),FIRST(α2), . . . ,FIRST(αn) are all pairwise
disjoint

2 If αi ⇒∗ ε then
FIRST(αj)

⋂
FOLLOW(A) = φ ,∀1 ≤ j ≤ n, i ̸= j.

If G is ε-free, condition 1 is sufficient.

LL(1) grammars

Provable facts about LL(1) grammars:
1 No left-recursive grammar is LL(1)

Direct case: A → Aα | β (useful productions).
Case 1: β is non-nullable. For any a ∈ FIRST(β ),

Aα ⇒ βα ⇒∗ aγα.

Thus a ∈ FIRST(Aα) too: condition 1 fails.
Case 2: β is nullable, so A is nullable.

If α is nullable, then ε ∈ FIRST(β )∩ FIRST(Aα): condition 1 fails.
Otherwise, for any a ∈ FIRST(α),

Aα ⇒∗ α ⇒∗ aγ ⇒ a ∈ FIRST(Aα),
A ⇒ Aα ⇒∗ Aaγ ⇒ a ∈ FOLLOW(A).

The same a triggers both alternatives: condition 2 fails.

Indirect case: expand the cycle to expose A ⇒+ Aα; the same case
analysis applies.

LL(1) grammars

Provable facts about LL(1) grammars:
2 No ambiguous grammar is LL(1)

Some string has two leftmost derivations. Where they first differ,
one A is expanded two ways and both still reach that string, so their
FIRST (or FIRST/FOLLOW) sets overlap. Hence LL(1) ⇒
unambiguous, though not conversely.

LL(1) grammars

Provable facts about LL(1) grammars:
3 Some languages have no LL(1) grammar

Some CFLs are inherently ambiguous i.e., no unambiguous CFGs
exist for that CFL.

4 A grammar which is not LL(1) may be converted into a LL(1)
grammar.

Consider S → aS | a. Not LL(1) since FIRST(aS) = FIRST(a). Use
left-factoring to get: S → aS′

S′ → aS′ | ε

accepts the same language and is LL(1)

Where we are

1 Context-free grammars

2 Top-down parsing
From scanning to parsing
Left recursion
Recursive descent and predictive parsing
FIRST, FOLLOW and LL(1) grammars
Building and using the LL(1) table
When LL(1) fails

LL(1) parse table construction

Input: Grammar G
Output: Parsing table M
Method:

1 ∀ productions A → α:
1 ∀a ∈ FIRST(α), add A → α to M[A,a]
2 If ε ∈ FIRST(α):

1 ∀b ∈ FOLLOW(A), add A → α to M[A,b]
2 If $ ∈ FOLLOW(A) then add A → α to M[A,$]

2 Set each undefined entry of M to error

If ∃M[A,a] with multiple entries then grammar is not LL(1).

Example

Our expression grammar:

1. S → E 6. T → FT ′

2. E → TE′ 7. T ′ →∗T
3. E′ →+E 8. | /T
4. | −E 9. | ε

5. | ε 10. F → num
11. | id

FIRST FOLLOW id num + − ∗ / $
S
E
E′

T
T ′

F
id
num
∗
/
+
−

Example: Calculating FIRST

1. S → E 6. T → FT ′

2. E → TE′ 7. T ′ →∗T
3. E′ →+E 8. | /T
4. | −E 9. | ε

5. | ε 10. F → num
11. | id

FIRST(E) ⊆ FIRST(S) From 1
FIRST(T) ⊆ FIRST(E) From 2
{+,−,ε} ⊆ FIRST(E′) From 3–5

FIRST(F) ⊆ FIRST(T) From 6
{∗,/,ε} ⊆ FIRST(T ′) From 7–9

{num,id} ⊆ FIRST(F) From 10, 11

Example: Calculating FIRST

1. S → E 6. T → FT ′

2. E → TE′ 7. T ′ →∗T
3. E′ →+E 8. | /T
4. | −E 9. | ε

5. | ε 10. F → num
11. | id

FIRST FOLLOW id num + − ∗ / $
S num,id
E num,id
E′ ε,+,−
T num,id
T ′ ε,∗,/
F num,id
id id −
num num −
∗ ∗ −
/ / −
+ + −
− − −

Example: Calculating FOLLOW

1. S → E 6. T → FT ′

2. E → TE′ 7. T ′ →∗T
3. E′ →+E 8. | /T
4. | −E 9. | ε

5. | ε 10. F → num
11. | id

{$} ⊆ FOLLOW(S) Start-symbol rule
FOLLOW(S) ⊆ FOLLOW(E) From 1

FIRST(E′)−{ε} ⊆ FOLLOW(T) From 2
FOLLOW(E) ⊆ FOLLOW(E′) From 2
FOLLOW(E) ⊆ FOLLOW(T) From 2
FOLLOW(E′) ⊆ FOLLOW(E) From 3, 4

FIRST(T ′)−{ε} ⊆ FOLLOW(F) From 6
FOLLOW(T) ⊆ FOLLOW(T ′) From 6
FOLLOW(T) ⊆ FOLLOW(F) From 6
FOLLOW(T ′) ⊆ FOLLOW(T) From 7, 8

Example: Calculating FOLLOW

1. S → E 6. T → FT ′

2. E → TE′ 7. T ′ →∗T
3. E′ →+E 8. | /T
4. | −E 9. | ε

5. | ε 10. F → num
11. | id

FIRST FOLLOW id num + − ∗ / $
S num,id $
E num,id $
E′ ε,+,− $
T num,id +,−,$
T ′ ε,∗,/ +,−,$
F num,id +,−,∗,/,$
id id −
num num −
∗ ∗ −
/ / −
+ + −
− − −

Example: Calculating the Parsing Table

1. S → E 6. T → FT ′

2. E → TE′ 7. T ′ →∗T
3. E′ →+E 8. | /T
4. | −E 9. | ε

5. | ε 10. F → num
11. | id

FIRST FOLLOW id num + − ∗ / $
S num,id $ 1 1 − − − − −
E num,id $ 2 2 − − − − −
E′ ε,+,− $ − − 3 4 − − 5
T num,id +,−,$ 6 6 − − − − −
T ′ ε,∗,/ +,−,$ − − 9 9 7 8 9
F num,id +,−,∗,/,$ 11 10 − − − − −
id id −
num num −
∗ ∗ −
/ / −
+ + −
− − −

Table driven Predictive parsing
Input: A string w and a parsing table M for a grammar G
Output: If w is in L(G), a leftmost derivation of w; otherwise, indicate an

error
1 push $ onto the stack; push S onto the stack;
2 let a = first symbol(w);
3 X = stack.top();
4 while X ̸= $ do
5 if X == a then
6 stack.pop(); let a = next symbol(w);

7 else if X is a terminal then
8 error();

9 else if M[X,a] is an error entry then
10 error();

11 else if M[X,a] = X → Y1Y2 · · ·Yk then
12 output the production X → Y1Y2 · · ·Yk;
13 stack.pop();
14 push Yk,Yk−1, · · ·Y1 in that order;

15 X = stack.top();
KC Sivaramakrishnan (IIT Madras) CS3300 - Monsoon 2026 Monsoon 2026 74 / 81



Where we are

1 Context-free grammars

2 Top-down parsing
From scanning to parsing
Left recursion
Recursive descent and predictive parsing
FIRST, FOLLOW and LL(1) grammars
Building and using the LL(1) table
When LL(1) fails

A grammar that is not LL(1): the dangling else

⟨stmt⟩ ::= if ⟨expr⟩ then ⟨stmt⟩
| if ⟨expr⟩ then ⟨stmt⟩ else ⟨stmt⟩
| other

Left-factored: 1. ⟨stmt⟩ ::= if ⟨expr⟩ then ⟨stmt⟩ ⟨stmt′⟩
2. | other
3. ⟨stmt′⟩ ::= else ⟨stmt⟩
4. | ε

Constraints relevant to the conflict (others omitted):

FIRST(⟨stmt′⟩) = {else,ε} From 3, 4
$ ∈ FOLLOW(⟨stmt⟩) Start-symbol rule

FOLLOW(⟨stmt⟩) ⊆ FOLLOW(⟨stmt′⟩) From 1
FIRST(⟨stmt′⟩)−{ε} ⊆ FOLLOW(⟨stmt⟩) From 1

FOLLOW(⟨stmt′⟩) ⊆ FOLLOW(⟨stmt⟩) From 3

A grammar that is not LL(1): the dangling else

Left-factored: ⟨stmt⟩ ::= if ⟨expr⟩ then ⟨stmt⟩ ⟨stmt′⟩ | other
⟨stmt′⟩ ::= else ⟨stmt⟩ | ε

Picking the smallest set that can satisfy the constraints gives us:
FOLLOW(⟨stmt′⟩) = {else,$}

Given ⟨stmt′⟩ ⇒∗ ε, LL(1) grammar requires
FIRST(else⟨stmt⟩)∩ FOLLOW(⟨stmt′⟩) = /0.

But FIRST(else⟨stmt⟩)∩ FOLLOW(⟨stmt′⟩) = {else}

The parsing table entry for M[⟨stmt′⟩,else] will contain both:
⟨stmt′⟩ ::= else⟨stmt⟩
⟨stmt′⟩ ::= ε

Intuitively, prioritise ⟨stmt′⟩ ::= else⟨stmt⟩ to associate else with
closest then.

Another common example

Here is a typical example where a programming language fails to
be LL(1):

⟨stmt⟩ → ⟨assignment⟩ | ⟨call⟩ | ⟨other⟩
⟨assignment⟩ → ⟨id⟩= ⟨expr⟩

⟨call⟩ → ⟨id⟩(⟨expr-list⟩)
This grammar is not in a form that can be left factored. We must
first replace assignment and call by the right-hand sides of their
defining productions:

⟨stmt⟩ → ⟨id⟩= ⟨expr⟩ | ⟨id⟩(⟨expr-list⟩) | ⟨other⟩

We left factor:

⟨stmt⟩ → ⟨id⟩⟨stmt’⟩ | ⟨other⟩
⟨stmt’⟩ → = ⟨expr⟩ | (⟨expr-list⟩)

See how the grammar obscures the language semantics.
Most of PL syntax cannot be expressed naturally as LL(1) grammar.

Error recovery in predictive parsing

An error is detected on a terminal mismatch or when M[A,a] is
empty.

Panic mode: discard input until a synchronizing token is reached.

If a ∈ FOLLOW(A), then a can legally come after A. Report a missing
or malformed A, pop A, and do not consume a.

Recovery example: id * + id

After matching *, the top of the stack is T, and the lookahead is +:

Stack : $ E’ T
Input : + id $

The parser consults M[T,+], which is an error entry. The error is
detected at +.
Since + ∈ FOLLOW(T), report “expected an operand after *”, pop T,
and continue at +.

The input remains invalid; recovery only lets parsing continue.

Parsing: key takeaways

Meaning: A CFG defines derivations and parse trees; ambiguity
gives one string more than one parse tree.
Strategy: Recursive descent may backtrack over alternatives;
predictive parsing chooses using lookahead.
Preparation: Grammar transformations can remove cycles,
ε-productions, and left recursion, or factor common prefixes—but
not every CFL has an LL(1) grammar.
Mechanics: FIRST and FOLLOW build the LL(1) table. The stack
and lookahead drive parsing; error entries detect mistakes, and
FOLLOW sets provide synchronization points for recovery.

KC Sivaramakrishnan (IIT Madras) CS3300 - Monsoon 2026 Monsoon 2026 81 / 81


	Context-free grammars
	Grammars and notation
	Derivations and parse trees
	Ambiguity

	Top-down parsing
	From scanning to parsing
	Left recursion
	Recursive descent and predictive parsing
	FIRST, FOLLOW and LL(1) grammars
	Building and using the LL(1) table
	When LL(1) fails


